
Häufig enthalten Excel-Dateien vertrauliche Daten. Da kommt es auf einen effektiven Schutz an. Leider entsprechen die Standardmechanismen von Excel nicht gerade den Erwartungen. Sie können Excel-Dateien zwar mit einem Kennwort schützen. Der Standardschutz lässt sich aber mit einfachen Mitteln in kürzester Zeit knacken. Hinzu kommt, dass viele Informationen im Klartext lesbar sind, wenn man eine kennwortgeschützte Excel-Datei in einem simplen Texteditor öffnet.

Doch Excel kann auch anders. Seit Excel 2002/XP können Sie Arbeitsmappen nicht nur mit einem Kennwortschutz versehen, Sie können die Mappen gleichzeitig auch noch verschlüsseln. So schalten Sie auf jeden Fall die Möglichkeit aus, geschützte Tabellendaten in einem Texteditor sichtbar zu machen. Außerdem sind die aktuellen Verschlüsselungsmethoden deutlich schwerer zu knacken. Als hundertprozentig sicher gelten zwar auch die neuesten Schutzmechanismen von Excel nicht. Angriffen von "Gelegenheits-Kriminellen" oder Amateur-Hackern schieben Sie damit aber durchaus einen Riegel vor.

Problematisch und zeitaufwendig könnte sich jetzt nur noch die Umstellung auf die neuesten Sicherheitsfunktionen gestalten. Jede schützenswerte Excel-Datei zu öffnen und manuell mit einem sicheren Verschlüsselungsalgorithmus zu versehen, kostet wertvolle Arbeitszeit.

Aus diesem Grund haben wir ein Makro entwickelt, das den Aufwand auf ein Minimum reduziert. Sie müssen nur noch die Dateien auswählen, die Sie per Kennwort schützen und verschlüsseln wollen, und das gewünschte Kennwort eingeben. Den Rest erledigt das Makro automatisch für Sie. Einziges Manko: Alle ausgewählten Dateien erhalten dasselbe Kennwort. Aber wenn Sie ein "starkes" Kennwort mit einer Länge von mindestens 10 Zeichen und einer Mixtur aus Buchstaben, Zahlen und Sonderzeichen wählen, sollte diese Einschränkung kein Problem sein.

Das Makro legen Sie in einer neuen Arbeitsmappe mit folgenden Schritten an:

· Öffnen Sie eine neue, leere Excel-Arbeitsmappe.

· Drücken Sie Alt+F8, um das Dialogfenster Makro einzublenden.

· Geben Sie im Feld Makroname folgende Bezeichnung ein:

Dateischutz
· Klicken Sie auf die Schaltfläche Erstellen.

· Sie gelangen in den Visual Basic-Editor, in dem ein neues Codefenster mit dem Gerüst der Prozedur "Dateischutz" angezeigt wird. Füllen Sie die Prozedur mit den erforderlichen Programmzeilen, so dass Sie schließlich folgendermaßen aussieht:

Sub Dateischutz()
Dim strDateifilter As String
Dim arrDateien
Dim strProvider As String
Dim strAlgorithmus As String
Dim lngKeyLaenge As Long
Dim PwdAlt, PwdNeu
Dim i As Long
Dim objMappe As Workbook
Dim strFehler As String

  On Error GoTo Schutzproc_Fehler
  If Val(Application.Version) < 12 Then
    strDateifilter = "Excel-Dateien (*.xls),*.xls"
  Else
    strDateifilter = "Excel-Dateien (*.xls?),*.xls?"
  End If
  arrDateien = Application.GetOpenFilename _
      (FileFilter:=strDateifilter, _
       Title:="Zu verschlüsselnde Dateien auswählen", _
       MultiSelect:=True)
  If IsArray(arrDateien) Then
    Select Case MsgBox _
        (Prompt:="Wollen Sie eine starke " & _
         "Verschlüsselungsmethode anwenden?" & _
         vbCr & "(Eventuell nicht kompatibel " & _
         "mit Excel 97/2000!)", _
         Buttons:=vbYesNoCancel, _
         Title:="Verschlüsselungsmethode")
      Case vbYes
        strProvider = _
          "Microsoft Strong Cryptographic Provider"
        strAlgorithmus = "RC4"
        lngKeyLaenge = 128
      Case vbNo
        strProvider = _
          "Office"
        strAlgorithmus = "OfficeStandard"
        lngKeyLaenge = 40
      Case vbCancel
        Exit Sub
    End Select

    PwdAlt = Application.InputBox _
        (Prompt:="Geben Sie das bislang " & _
                 "benutzte Kennwort ein:", _
         Title:="Altes Kennwort", _
         Type:=2)
    If PwdAlt = False Then Exit Sub
    PwdNeu = Application.InputBox _
        (Prompt:="Geben Sie ein " & _
                 "neues Kennwort ein:", _
         Title:="Neues Kennwort", _
         Type:=2)
    If PwdNeu = False _
        Or PwdNeu = "" Then Exit Sub

    On Error Resume Next
    Application.ScreenUpdating = False
    For i = 1 To UBound(arrDateien)
      Set objMappe = Application.Workbooks.Open _
          (Filename:=arrDateien(i), _
           Password:=PwdAlt)
      If objMappe Is Nothing Then
        strFehler = strFehler & _
            arrDateien(i) & vbCr
      Else
        Err.Clear
        With objMappe
          .SetPasswordEncryptionOptions _
              PasswordEncryptionProvider:= _
                strProvider, _
              PasswordEncryptionAlgorithm:= _
                strAlgorithmus, _
              PasswordEncryptionKeyLength:= _
                lngKeyLaenge
          .Password = PwdNeu
          If Err <> 0 Then
            strFehler = strFehler & _
                arrDateien(i) & vbCr
            Err.Clear
            .Close SaveChanges:=False
          Else
            .Close SaveChanges:=True
          End If
        End With
      End If
    Next i
  End If

Schutzproc_Ende:
  If strFehler <> "" Then
    MsgBox "Verschlüsselung abgeschlossen." & vbCr & _
      "Fehler/nicht bearbeitete Dateien:" & vbCr & _
      strFehler
  Else
    MsgBox "Verschlüsselung fehlerfrei abgeschlossen."
  End If
  Application.ScreenUpdating = True
  Exit Sub

Schutzproc_Fehler:
  strFehler = "Fehlernr.: " & Err.Number & vbCr & _
    Err.Description & vbCr & strFehler
  Resume Schutzproc_Ende
End Sub
· Beenden Sie den Visual Basic-Editor per Datei-Schließen und zurück zu Microsoft Excel.

· Speichern Sie die aktuelle Arbeitsmappe unter einem Namen Ihrer Wahl.

Die weiteren Details und Hintergrundinformationen liefern wir im zweiten Teil dieses Tipps nach.

Im ersten Teil dieses Tipps haben wir Ihnen ein Makro vorgestellt, mit dem Sie mehrere Dateien automatisch mit einem Kennwort versehen und gleichzeitig mit der höchstmöglichen Excel-Sicherheitsstufe verschlüsseln können. Erfahren Sie nun mehr zur Funktionsweise und zu den Hintergründen.

Wenn Sie nicht schon selbst mit dem Makro experimentiert haben, hier zunächst die Schritte zum Starten der Makrolösung. Aber Vorsicht! Testen Sie das Makro am besten erst mit Kopien Ihrer Originaldateien. Vor allem in Mehrbenutzerumgebungen oder in Konstellationen, in denen Dateien mit unterschiedlichen Excel-Versionen bearbeitet werden müssen, sollten Sie unbedingt prüfen, ob sich die geschützten Dateien anschließend reibungslos öffnen lassen - korrekte Kennworteingabe natürlich vorausgesetzt.

1. Schließen Sie möglichst alle geöffneten Excel-Dateien - oder zumindest die Dateien, die Sie von dem Makro mit einem Kennwort versehen lassen wollen.

2. Öffnen Sie die Arbeitsmappe mit dem "Dateischutz"-Makro, also die Datei, die Sie in Teil 1 dieses Tipps angelegt haben. Sorgen Sie dafür, dass die Ausführung von Makros aktiviert ist.

3. Drücken Sie Alt + F8.

4. In der Liste des Dialogfensters Makro markieren Sie den Makronamen "Dateischutz" und danach klicken Sie auf die Schaltfläche Ausführen.

5. Daraufhin erscheint als erstes das Dialogfenster zum Öffnen von Dateien, und zwar mit dem Titel Zu verschlüsselnde Dateien auswählen. Wechseln Sie wie gewohnt in den Ordner, in dem sich die Excel-Dateien befinden, die Sie schützen wollen. Und markieren Sie anschließend die gewünschten Dateinamen (mit Strg+A können Sie auch alle Dateien markieren). Zur Bestätigung klicken Sie auf die Schaltfläche Öffnen.

6. Als nächstes fragt Sie das Makro, ob Sie eine starke Verschlüsselungsmethode anwenden wollen. Klicken Sie auf Ja. - Nur wenn Sie darauf angewiesen sind, Excel-Dateien auch mit Excel 97 oder 2000 bearbeiten zu können, klicken Sie auf Nein. Dann wird das Makro mit dem herkömmlichen, unverschlüsselten Kennwortschutz von Excel arbeiten.

7. Für den Fall, dass die im vorletzten Schritt markierten Dateien bereits kennwortgeschützt sind (alle Dateien müssen dasselbe Kennwort verwenden), können Sie dieses alte Kennwort jetzt eingeben. Zur Bestätigung klicken Sie auf Ok. - Sollten die markierten Dateien bislang ungeschützt sein, lassen Sie das Feld leer und klicken gleich auf Ok.

8. Nun werden Sie erneut zur Kennworteingabe aufgefordert. Dieses Mal handelt es sich um das neue Kennwort, das auch zur Verschlüsselung dient. Eine Eingabe ist zwingend erforderlich. Wichtig ist zudem, dass Sie sich das Kennwort gut merken, denn sonst können Sie die ausgewählten Dateien später nicht mehr öffnen. Bestätigen Sie die Kennworteingabe wieder mit Ok.

9. Danach beginnt das Makro mit der Zuweisung des Kennwortschutzes. Je nach Anzahl und Größe der Dateien kann dieser Vorgang einige Zeit in Anspruch nehmen. Wenn alle Dateien verarbeitet sind, informiert Sie das Makro über die Fertigstellung.

Vielleicht fragen Sie sich, wie das Makro verschiedene Verschlüsselungsmethoden anwenden kann. Das ist keineswegs eine Funktion, die sich nur per VBA-Code einsetzen lässt. Zumindest in Excel 2003 und 2002/XP können Sie die Art des Dateischutzes auch über die Benutzeroberfläche einstellen: Wählen Sie Extras-Optionen an und wechseln Sie dann auf die Registerkarte Sicherheit. Dort klicken Sie neben dem Feld Kennwort zum Öffnen auf die Schaltfläche Erweitert. Daraufhin öffnet sich ein Dialogfenster, welches alle verfügbaren Verschlüsselungstypen auflistet. Die vorgestellte Makrolösung verwendet standardmäßig den Typ "Microsoft Strong Cryptographic Provider" mit der Standard-Schlüssellänge von 128 Bit. - Nur wenn Sie bei der Makroausführung aufgrund der Kompatibilität zu Excel 97/2000 die starke Verschlüsselungsmethode ablehnen, setzt das Makro den Verschlüsselungstyp "Office 97/2000-kompatibel" ein.

In Excel 2007 wurde der Zugriff auf individuelle Verschlüsselungstypen aus der Benutzeroberfläche entfernt. Dateien im "alten" Excel 97-2003-Format schützt Excel 2007 standardmäßig im unsicheren Verschlüsselungstyp "Office 97/2000-kompatibel". Nur bei neuen XLSX-Dateien wendet Excel 2007 den "Microsoft Enhanced RSA and AES Cryptographic Provider" an. Da dieser aber in älteren Excel-Versionen noch als "Prototype" ausgewiesen wird, sind Inkompatibilitäten nicht auszuschließen!

Umso sinnvoller ist der Einsatz der in diesem Tipp vorgestellten Makrolösung. Sie vereinheitlicht die Verschlüsselung für alle Excel-Versionen von 2007 bis 2002/XP. Und Sie können sicher sein, dass Ihre Dateien - egal ob XLS oder XLSX - effektiv geschützt sind.
